

graph twoway scatteri — Scatter with immediate arguments[Description](#)
[Options](#)[Quick start](#)
[Remarks and examples](#)[Menu](#)
[Also see](#)[Syntax](#)

Description

`scatteri` is an immediate version of `twoway scatter`; see [\[U\] 19 Immediate commands](#) and [\[G-2\] graph twoway scatter](#). `scatteri` is intended for programmer use but can be useful interactively.

Quick start

An immediate scatterplot with a single point at $x = 1$ and $y = 2$

```
twoway scatteri 2 1
```

Add label “Note this point” at 6 o’clock

```
twoway scatteri 2 1 (6) "Note this point"
```

Add a point at (3,1) with label “This point too” at 12 o’clock

```
twoway scatteri 2 1 (6) "Note this point" 1 3 (12) "This point too"
```

Highlight the point at (15,22) on a scatterplot of y versus x

```
twoway scatter y x || scatteri 22 15 (2) "Note this point"
```

Same as above, but show only the label

```
twoway scatter y x || scatteri 22 15 (2) "Note this point", ///  
msymbol(none)
```

Menu

Graphics > Twoway graph (scatter, line, etc.)

Syntax

```
twoway scatteri immediate_values [, options]
```

where *immediate_values* is one or more of

```
#y #x [ (#clockposstyle) ] [ "text for label" ]
```

See [G-4] *clockposstyle* for a description of *#clockposstyle*.

Options

options are as defined in [G-2] **graph twoway scatter**, with the following modifications:

If "*text for label*" is specified among any of the immediate arguments, option `mlabel()` is assumed.

If (*#clockposstyle*) is specified among any of the immediate arguments, option `mlabvposition()` is assumed.

Remarks and examples

[stata.com](http://www.stata.com)

Immediate commands are commands that obtain data from numbers typed as arguments. Typing

```
. twoway scatteri 1 1 2 2, any_options
```

produces the same graph as typing

```
. clear
. input y x
      y      x
1. 1 1
2. 2 2
3. end
. twoway scatter y x, any_options
```

`twoway scatteri` does not modify the data in memory.


`scatteri` is intended for programmer use but can be used interactively. In

[G-3] *added_text_options*, we demonstrated the use of option `text()` to add text to a graph:

```
. twoway qfitci mpg weight, stdf ||
scatter mpg weight, ms(0) ||
text(41 2040 "VW Diesel", place(e))
text(28 3260 "Plymouth Arrow", place(e))
text(35 2050 "Datsun 210 and Subaru", place(e))
```

Below we use `scatteri` to obtain similar results:

```
. twoway qfitci mpg weight, stdf ||
scatter mpg weight, ms(0) ||
scatteri 41 2040 (3) "VW Diesel"
        28 3260 (3) "Plymouth Arrow"
        35 2050 (3) "Datsun 210 and Subaru"
        , msymbol(i) legend(order(1 2 3))
```



We translated `text(..., place(e))` to `(3)`, 3 o'clock being the *clockposstyle* notation for the east *compassdirstyle*. Because labels are by default positioned at 3 o'clock, we could omit `(3)` altogether:

```
. twoway qfitci mpg weight, stdf ||
scatter mpg weight, ms(0) ||
scatteri 41 2040 "VW Diesel"
        28 3260 "Plymouth Arrow"
        35 2050 "Datsun 210 and Subaru"
        , msymbol(i) legend(order(1 2 3))
```

We specified the `msymbol(i)` option to suppress displaying the marker symbol. We also specified the `legend(order(1 2 3))` option to include only the keys from the first three plots in the legend.

□ Technical note

Programmers: Note carefully `scatter`'s *advanced_option* `recast()`; see [G-3] *advanced_options*. It can be used to good effect, such as using `scatteri` to add areas, bars, spikes, and dropped lines. □

Also see

[G-2] [graph twoway scatter](#) — Twoway scatterplots

[U] 19 [Immediate commands](#)

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2023 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the FAQ on [citing Stata documentation](#).